

Identifying Entries And Exits

Smoothing The Bollinger %b

This variation of the popular Bollinger %b indicator gives clear turning points that will help traders identify entry and exit points while prices are moving between the Bollinger bands.

by Sylvain Vervoort

By now, everybody in the business knows that Bollinger bands were developed by John Bollinger. What they actually are is probably equally known. The Bollinger bands in Figure 1 consist of a set of three curves drawn in relation to price data. The middle band is usually a simple 20-bar moving average that serves as the base for the upper and lower bands.

Upper band = Middle band + 2 x 20-period closing prices standard deviation

Lower band = Middle band – 2 x 20-period closing prices standard deviation

Most technical analysis programs have a built-in function for displaying Bollinger bands. Here is the MetaStock formula for Bollinger bands using basic functions:

```
{The middle band}
Mov(Close,20,Simple);

{The upper band}
2*Stdev(Close,20 ) + Mov(Close,20,Simple);

{The lower band}
Mov(Close,20,Simple)-2*Stdev(Close,20)
```

In a normal distribution, most prices of a set of data are close to the average, with relatively few prices moving in one extreme or the other. If you were to look at a normally distributed data on a graph, it would look like a bell curve. Not all price data will reflect a perfect curve, of course. The standard deviation of the 20-period closing prices is a statistic that tells you how tight the price data are clustered around the mean.

The standard deviation of the moving average is a method used to measure price volatility. With trending prices, the bands will be wider as a result of the higher volatility in price, mov-

CHRISTINE MORRISON

ing farther away from the mean, whereas during consolidation periods, bands tend to be narrower as a result of smaller price moves closer to the mean. This changing bandwidth is used for volatility-based trading opportunities.

%b is a measure of where prices are in relation to the outer Bollinger bands and therefore strongly related to volatility. %b was created as an oscillator to show overbought and oversold situations when price is moving close to or beyond the upper or lower Bollinger bands. The basic %b formula is:

$$\%b = \frac{(\text{Close} - \text{Lower band})}{(\text{Upper band} - \text{Lower band})}$$



FIGURE 1: BOLLINGER BANDS. Bollinger bands consist of three curves drawn in relation to price data. The middle band is usually a 20-bar simple moving average that serves as the base for the upper and lower bands.

For the basic formula, you multiply the %b result with 100 to get an oscillator moving between zero and 100 (with overshoots):

```
{The upper band}
upper:= 2*Stdev(CLOSE,20) + Mov(CLOSE,20,SIMPLE);

{The lower band}
lower:=Mov(CLOSE,20,SIMPLE)-2*Stdev( CLOSE,20);

{%b}
percb:=(C-lower)/(upper-lower)*100;
percb
```

This formula can be simplified by applying some math. Here is the following result for the basic (MetaStock) formula for a Bollinger band %b indicator using closing prices and a simple moving average:

```
{BB%b_C_S}
percb:=(C+2*Stdev(C,20)-Mov(C,20,S))/
(4*Stdev(C,20))*100;
percb
```

This is the indicator you see at the bottom of the chart of GAIAM Inc. (GAIA) in Figure 2. The top chart displays the price chart overlaid with the Bollinger bands. As you can see, the indicator is leading most of the time, showing high levels, low levels, and eventually divergences prior to turning points in price.

Unfortunately, it is a very choppy oscillator. Finding the more important turning points and trying to trade with the %b indicator, day by day, is not an easy task.

Normal and hidden divergent moves make this indicator an ideal tool to help find entry and exit points.

SMOOTHING IT AND REDUCING DELAYS

I have already discussed my own version of this indicator in my book *Capturing Profit With Technical Analysis*. This time, I am introducing you to a more usable version of this indicator. In the second article of this series, I will show you a trading system based on the leading property of this indicator. Right now, however, let us first look for ways to smooth this indicator.



FIGURE 2: BOLLINGER BANDS AND %B INDICATOR. The indicator is leading most of the time, showing high levels, low levels, and eventually divergences prior to turning points in price.



FIGURE 3: BASIC BOLLINGER SMOOTHING STEPS. Here you see the results of each of the four steps used to smooth the Bollinger bands %b indicator.

In Figure 3, you can see the price chart of GAIA at the top and the basic closing prices %b indicator (BB%b_C) displayed on the bottom subchart.

Step 1:

A good basic smoothing technique that creates a very small delay is the heikin-ashi recalculated prices. The heikin-ashi is a Japanese charting technique, detailed in a STOCKS & COMMODITIES article in February 2004. I am using my own average heikin-ashi closing price "haC," which is calculated as follows:

```
haOpen:=(Ref((O+H+L+C)/4,-1) + PREV)/2;
haC:=((O+H+L+C)/4+haOpen+Max(H,haOpen)+Min(L,haO
pen))/4;
```

You can use this haC closing price instead of the normal closing price in the previous basic formula:

```
(haC+2*Stdev(haC,20)-Mov(haC,20,SIMPLE))/
(4*Stdev(haC,20))*100;
```

Compare this modified version called "BB%b_step1" in Figure 3, with the basic version at the bottom of this chart. As you can see, there is some amount of smoothing already, but not enough to make your trading decisions that much simpler. If you look closely, you may see a small amount of lagging. This means that if you want more smoothing, you will have to limit the delay as much as you can.

Step 2:

You can achieve more smoothing with only a little lag by using the triple exponential moving average (TEMA), which was explained in the February 1994 STOCKS & COMMODITIES. The data is sent several times through the same filter and combined afterward:

```
TEMA = (3*EMA–3*EMA(EMA)) + EMA(EMA(EMA))
```

EMA stands for exponential moving average.

MetaStock has a built-in TEMA function. However, just using TEMA on the heikin-ashi recalculated prices still creates more lag than I would like. To limit the lag as much as possible, I will also use a zero-lagging technique. The principle for limiting the lag was analyzed in the January 2000 STOCKS & COMMODITIES. In this second step, I smooth with an eight-period TEMA:

```
haOpen:=(Ref((O+H+L+C)/4,-1) + PREV)/2;
haC:=((O+H+L+C)/4+haOpen+Max(H,haOpen)+Min(L,haOpen))/4;
TMA1:= Tema(haC,8);
TMA2:= Tema(TMA1,8);
Diff:= TMA1 - TMA2;
ZlHA:= TMA1 + Diff;
percb:=(ZLHA+2*Stdev(ZLHA,20)-Mov(ZLHA,20,S))/
(4*Stdev(ZLHA,20))*100;
percb
```

You can see the result in Figure 3 displayed as "BB%b_step2."

Step 3:

So far, so good. However, I would like to see more smoothing and, of course, minimal or no lag. What I can do is smooth the standard deviation bands with TEMA. And instead of using the simple moving average for the referenced middle band, I can use a faster exponential moving average (EMA) or a weighted moving average (WMA). I prefer the WMA because it tends to move more smoothly than the EMA.

I can also bring down the averaging period as long as the smoothing is acceptable. This is why I use an 18-period average instead of the standard 20-period average in the next formula.

Introducing these additional changes results in the following:

```
haOpen:=(Ref((O+H+L+C)/4,-1) + PREV)/2;
haC:=((O+H+L+C)/4+haOpen+Max(H,haOpen)+Min(L,haO
pen))/4;
TMA1:= Tema(haC,8);
TMA2:= Tema(TMA1,8);
Diff:= TMA1 - TMA2;
ZlHA:= TMA1 + Diff;
percb:=(Tema(ZLHA,8)+2*Stdev(Tema(ZLH
A,8),18)-Mov(Tema(ZLHA,8),18,WEIGHTED))/
(4*Stdev(Tema(ZLHA,8),18))*100;
percb
```

In Figure 3, you can compare the result of BB%b_step3 with the basic BB%b_C at the bottom of the chart. You have to

agree that it looks a lot less choppy and that delays are most likely not a problem, especially because the %b is a leading indicator and tends to usually be fast.

Step 4:

In this, the final step, I make the WMA and the TEMA adjustable and instead of using fixed reference lines at, for example, zero, 50, and 100, I use standard deviation lines with 1.6 standard deviations as the default over a look-back period of 63 days. The final SVE_BB%b code for MetaStock can be found in sidebar "MetaStock Code For SVE_BB%b."

APPLYING THE INDICATOR

Using the chart of GAIA as an example, I will show you how basically to use this indicator. Keep in mind I will introduce a trading system using this indicator in part 2 of this series.

I have split the chart of GAIA into two parts. Figure 4 is the first part ended March 9, 2009. In November there was a big price drop and some recuperation. After that, price started moving further down in a very gentle slope. What does the SVE_BB%b indicator tell you about that period?

In Figure 4, you can see the big price drop in November 2008 as well as the reaction to the drop, which went on until the beginning of December 2008. This was a convergent move with lower tops in price and the indicator. During phase 1, the price continued its down move, making a lower top in price. But the indicator makes a higher top at the beginning of January 2009. This is an inverse or hidden divergence pointing to a continuation of the previous downtrend.



FIGURE 4: PRICE DROP. In November, there was a big price drop and some recuperation. After that, price started moving further downward with a very gentle slope. Note the convergences and divergences between the price and indicator.

The SVE_BB% is often a leading indicator with clear turning points.

Prices continue on their downward move until the middle of February 2009. In phase 2, you see a similar situation as in phase 1, where there are lower tops in price and higher tops in the indicator. This is another hidden divergence telling you that price will continue its down move. In phase 3, price is making lower lows, but now the indicator is making higher lows. This is a normal divergence telling you that you should expect the price to move up.

Note the dojis in the candlestick chart where price is bottoming. In addition, note that the price has been moving within narrow Bollinger bands, indicating low volatility for more than three months already.

Clearly, you can open a position here with a very good risk-to-reward ratio, with a buy at $2.53 and an initial closing price stop at $2.38. This is an ideal entry for opening a long position with a risk of only 6%.

Let's continue with Figure 5, the second part of the chart of GAIA. Phase 3 actually brought a trend reversal and started a new price up move. There is a nice convergent up move until early May 2009. At phase 4, you see a negative divergence with higher tops in price but lower tops in the indicator. This suggests an upcoming price reversal.

Seems like a good time to take the more than 100% profit! What follows is a correction back to the lower Bollinger band. The end of this correction (phase 5) is showing a positive divergence with lower lows in price and higher lows in the indicator. This pushes prices up again. You could go for a new long position here. The tops in price and indicator at the beginning of June, or end of phase 6, show a negative divergence, which pushes prices down for a bigger correction. It now looks like prices have reached the end of this correction and there is a new positive divergence in phase 7.

METASTOCK CODE FOR SVE_BB%b

```
period:=Input("%b period: ",1,100,18);
TeAv:=Input("Tema average: ",1,30,8);
afwh:= Input("Standard deviation high ",.1,5,1.6);
afwl:= Input("Standard deviation Low ",.1,5,1.6);
afwper:= Input("Standard deviation period ",1,200,63);
haOpen:=(Ref((O+H+L+C)/4,-1) + PREV)/2;
haC:=((O+H+L+C)/4+haOpen+Max(H,haOpen)+Min(L,haOp
   en))/4;
TMA1:= Tema(haC,TeAv);
TMA2:= Tema(TMA1,TeAv);
Diff:= TMA1 - TMA2;
ZlHA:= TMA1 + Diff;
percb:=(Tema(ZLHA,TeAv)+2*Stdev(Tema(ZLHA,TeAv),period)-
   Mov(Tema(ZLHA,TeAv),period,WEIGHTED))/(4*Stdev(Tema(
   ZLHA,TeAv),period))*100;
percb;
50+afwh*Stdev(percb,afwper);
50-afwl*Stdev(percb,afwper);
50
```



FIGURE 5: THE UP MOVE. Here you see a new price up move. There is a nice convergent up move until early May 2009. After that, you see a negative divergence between price and indicator suggesting a trend reversal. Note the relationship between price moves and the indicator during the entire correction.

Interestingly, at the beginning of October 2009, the price of GAIA went up to about $8.

CONCLUSION

The SVE_BB%b is often a leading indicator making smooth moves with clear turning points. Normal and hidden divergent moves make it an ideal tool to help find entry and exit points while watching price moving between the Bollinger bands. I find it easier to use than the original %b indicator.

In the second article of this series, I will introduce a good trading system using the SVE_BB%b indicator and include some ideas together with other technical analysis tools. Stay tuned!

Belgium-based Sylvain Vervoort is a retired electronics engineer who has been using technical analysis for more than 30 years. His book, Capturing Profit With Technical Analysis, *published by Marketplace Books, is now available. Vervoort may be reached at sve.vervoort@scarlet.be or via his website, http://stocata.org.*

SUGGESTED READING

Bollinger, John [2001]. *Bollinger On Bollinger Bands*, McGraw-Hill.

Mulloy, Patrick [1994]. "Smoothing Data With Less Lag," *Technical Analysis of* STOCKS & COMMODITIES, Volume 12: February.

Sharp, Joe [2000]. "More Responsive Moving Averages," *Technical Analysis of* STOCKS & COMMODITIES, Volume 18: January.

Valcu, Dan [2004]. "Using The Heikin-Ashi Technique," *Technical Analysis of* STOCKS & COMMODITIES, Volume 22: February.

Vervoort, Sylvain [2008]. "Trading Medium-Term Divergences," *Technical Analysis of* STOCKS & COMMODITIES, Volume 26: February.

_____ [2009]. *Capturing Profit With Technical Analysis*, Marketplace Books.

S&C